PICAXE SAMPLE PROGRAMS

The following Picaxe programs open in Notepad. 

These are samples of the program that we used in the OARC CLUB PROJECT

These programs use ordinary hardware such as displays. Keyboards and an interface program to a PC (RealTerm).  All of the software that was used in the project is free.

Note: To open a file please right-click on each icon for the appropriate file and then select “package objects” and then “activate contents”. You will receive a warning indicating that the “package that you are about to open is a file type that can be unsafe”. Since none of the files included are executable files please ignore the warning and click on OK and the file will open in Notepad.

_1306485086/PICAXEPOWERMETER.BAS
; PicaxePower Meter



; program flow

; set up i2c slave for display

;  adc into b0

; read power from EEPROM

; use bintoascii to separate the power numbers

; write to the display

; loop

; can add keyboard to answer questions about averaging



pause 500

main:

readadc 3,b0

read b0,b1	;read from eeprom,put nr. in b1

bintoascii b1,b2,b3,b4



debug



if b2 =48 then

gosub display1

else

gosub display2

endif



goto main



display2:

i2cslave $C6,i2cslow,i2cbyte

pause 10

;writei2c 0,(254,1,255)

pause 30

writei2c 0,(254,128,255)

pause 10

writei2c 0,("Power ",b2,b3,".",b4," Watts",255)

return







display1:

i2cslave $C6,i2cslow,i2cbyte

pause 10

;writei2c 0,(254,1,255)

pause 30

writei2c 0,(254,128,255)

pause 10

writei2c 0,("Power  ", b3,".",b4," Watts",255)

return





EEPROM $0,(0, 0 ,0 ,0 ,0 ,0 ,0 ,0 ,0 ,0 ,0 ,0 ,0 ,0 ,0 ,0)


EEPROM $10,(0, 0 ,0 ,1 ,1 ,1 ,1 ,1 ,1 ,1 ,1 ,1 ,1 ,1 ,1 ,1)


EEPROM $20,(2, 2 ,2 ,2 ,2 ,2 ,2 ,2 ,2 ,3 ,3 ,3 ,3 ,3 ,3 ,3)


EEPROM $30,(4, 4 ,4 ,4 ,4 ,4 ,4 ,5 ,5 ,5 ,5 ,5 ,6 ,6 ,6 ,6)


EEPROM $40,(6, 6 ,7 ,7 ,7 ,7 ,8 ,8 ,8 ,8 ,8 ,9 ,9 ,9 ,9 ,10)


EEPROM $50,(10, 10 ,10 ,11 ,11 ,11 ,11 ,12 ,12 ,12 ,12 ,13 ,13 ,13 ,14 ,14)


EEPROM $60,(14, 14 ,15 ,15 ,15 ,16 ,16 ,16 ,17 ,17 ,17 ,18 ,18 ,18 ,19 ,19)


EEPROM $70,(19, 20 ,20 ,20 ,21 ,21 ,21 ,22 ,22 ,23 ,23 ,23 ,24 ,24 ,24 ,25)


EEPROM $80,(25, 26 ,26 ,26 ,27 ,27 ,28 ,28 ,28 ,29 ,29 ,30 ,30 ,31 ,31 ,31)


EEPROM $90,(32, 32 ,33 ,33 ,34 ,34 ,35 ,35 ,36 ,36 ,36 ,37 ,37 ,38 ,38 ,39)


EEPROM $A0,(39, 40 ,40 ,41 ,41 ,42 ,42 ,43 ,43 ,44 ,44 ,45 ,45 ,46 ,47 ,47)


EEPROM $B0,(48, 48 ,49 ,49 ,50 ,50 ,51 ,52 ,52 ,53 ,53 ,54 ,54 ,55 ,56 ,56)


EEPROM $C0,(57, 57 ,58 ,58 ,59 ,60 ,60 ,61 ,62 ,62 ,63 ,63 ,64 ,65 ,65 ,66)


EEPROM $D0,(67, 67 ,68 ,68 ,69 ,70 ,70 ,71 ,72 ,72 ,73 ,74 ,74 ,75 ,76 ,76)


EEPROM $E0,(77, 78 ,79 ,79 ,80 ,81 ,81 ,82 ,83 ,83 ,84 ,85 ,86 ,86 ,87 ,88)


EEPROM $F0,(89, 89 ,90 ,91 ,92 ,92 ,93 ,94 ,95 ,95 ,96 ,97 ,98 ,98 ,99 ,100)





















_1306485091/SImplePauseTest.bas
' Pause Test

' use a variable bo  [Byte]

'

' max value =255

' try 256





'	Initialize Variables

b0 = 255	'set value of b0



''''''''''''''''''''''''''''''''''''

'	Main Program 



''''''''''''''''''''''''''''''''''''

main:		 ' a label

high 0

pause b0	'milliseconds

low 0

pause b0

goto main	'go to main



'''''''''''''''''''''''''''''''''''






_1306485501/KEYINEEPROMTABLE.BAS
' Example to show how to convert computer 

' keyboard codes into ASCII values

' For  PICAXE-28X1

' 

' i2c Display used



' The ASCII values are preloaded into the data

' memory by use of the EEPROM command, When a

' key hit is detected it's value is used as 

' the data memory address with the read command.

' This enables the correct ASCII value to be

' read back from the data memory into variable b1







; set address for display here

i2cslave $C6,i2cslow,i2cbyte	      ;set up i2cslave for lcd

writei2c 0,(254,1,255)		 	;clear display

pause 30					; wait for display to clear



								

main:



							'get the keyboard hit  

kbin[1000,main],b0				'go to main after 1000 ms time out. gets b1

						

							'convert key value from keyin command

							'into an ASCII character in b1

read b0,b1 ;read from eeprom into b1	'read from EEPROM in Picaxe Chip

							'display ASCII value on display

;

writei2c 0,(254,128,255)		

writei2c 0, (b1,255)	

debug

goto main

end





				

; Compiler is not case sensitive         



; my ascii table

; 

; # symbol means that I have not assigned the key



 EEPROM $00,("################")		

 EEPROM $10,("#####Q1???ZSAW2#")		

 EEPROM $20,("#CXDE43###VFTR5#")

 EEPROM $30,("#NBHGY6???MJU78#")

 EEPROM $40,("#,KIO098?./L;P##")

 EEPROM $50,("################")	

 EEPROM $60,("################")		

 EEPROM $70,("################")

 EEPROM $80,("################") 





		





;Scan Codes from keyboard...hex



;	A	1C		0	45		ctrl			14

;	B	32		1	16		Shift			12

;	C	21		2	1E		spacebar		29

;	D	23		3	26		up arrow		75

;	E	24		4	25		down arrow		72

;	F	2B		5	2E		right arrow		74

;	G	34		6	36		left arrow		6B

;	H	33		7	3D		period		49

;	I	43		8	3E		comma			41

;	J	3B		9	46		forward slash	4A

;	K	42		F1	05		BackSpace		66

;	L	4B		F2	06		Enter			5A

;	M	3A		F3	04		Esc			76

;							semi-colon		4C

;	N	31		F4	0C

;	O	44		F5	03

;	P	4D		F6	0B

;	Q	15		F7	83

;	R	2D		F8	0A

;	S	1B		F9	01

;	T	2C		F10	09

;	U	3C		F11	78

;	V	2A		F12	07

;	W	1D

;	X	22

;	Y	35

;	Z	1A	
























_1306934939/IAMBICKEYER_SpeedCntrl0.BAS
;

;			Iambic Keyer

;

;Ottawa Amateur Radio club project Feb 2008

;

;Prepared by VE3TLY and VE3KL

;

;This program generates Morse Code using a PICAXE 28X1 microcontroller.

;	- it uses a paddle for input, and

;	- it will directly drive a transceiver keyer input.

;

;Code speed is controlled by a pot which is connected between 5V and ground.

;The pot slider is connected to "ADC0" [PICAXE chip pin-2 <==> Stack-28 Pin-16]

;

;Paddle connected as:

; 	dot:  "In0" [PICAXE chip pin-11 <==> Stack-28 Pin-8]

;	dash: "In1" [PICAXE chip pin-12 <==> Stack-28 Pin-9]

;

;Keyer output appears on "Out0" [PICAXE chip pin-21 <==> Stack-28 Pin-0]

;

;	Morse speed:

;	dot_length=1200/WPM   in ms. WPM is words per minute

;

;

;Set Variables

	symbol dot_length=w0	;length of dot  [ms]

	symbol dash_length=w1	;length of dash [ms]

	symbol char_space	=w2	;length of space between characters [ms] 

	symbol letter_space =w3	;space between letters [ms]

	symbol word_space=w4	;space between words [ms]

;

;	initialize variables

	dot_length=60

;

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

main:

	readadc 0, dot_length		;the "ADC0" value sets the basic delay "dot_length".

      dash_length=3*dot_length	;all other delays are scaled using "dot_length"

	char_space = dot_length

	letter_space=2*dot_length

	word_space=6*dot_length

	

	if pin0=0 and pin1 =1 then

	gosub dot

	endif



	if pin1=0 and pin0 =1 then

	gosub dash

	endif



	if pin0=0 and pin1=0 then

	gosub dotdash

	endif



goto main





;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

;	Define subroutines

;



dot:

	high 0

	pause dot_length

	low 0

	pause char_space

	return



dash:

	high 0

	pause dash_length

	low 0

	pause char_space

	return



dotdash:

	high 0

	pause dot_length

	low 0

	pause char_space

	high 0

	pause dash_length

	low 0

	pause char_space

	return







	




















_1306485094/OARC1_PAUSETEST_SUBS.BAS
' Pause Test using subroutines

' Lights diodes three times

' 

'

' uses word variables  w0,w1

' 



''''''''''''''''''''''''''''''''''''''



w0 = 500	'Initialize variabless

w1=1000

'''''''''''''''''''''''''''''''''''''



pause 500	'Main Program

gosub Green

pause 500

gosub Red

w0=2000

pause 500

gosub Green

end			'End needed so Green and Red routines are executed accidently



'''''''''''''''''''''''''''''''''''''''''''''''''



Green:   'Light green LED Subroutine

high 1

pause w0

low 1

pause w0

return	



''''''''''''''''''''''''''''''''''''''''''''''



Red:		'Light Red LED  Subroutine

high 0

pause w1

low 0

pause w1

return



''''''''''''''''''''''''''''''''''''''''''''''






_1306485466/OARC2_PAUSETEST_LOOP.BAS
' Pause Test using subroutines and a do loop

' use word variables w0,w1

'

' 

'

''''''''''''''''''''''''''''''''''''''''''''''''

	'Initialize variabless

w0=500

w1=50

''''''''''''''''''''''''''''''''''''''''''''''

	'Main Program

do

w0=w0 +500

pause 500

gosub Green

gosub Red

loop while w0 < 3000

end

	



'''''''''''''''''''''''''''''''''''''''''''''''''



Green:   'Light green LED Subroutine

high 1

pause w0

low 1

pause w0

return	



Red:		'Light Red LED  Subroutine

high 0

pause w1

low 0

pause w1

return










_1306485097/OARC_DIGITALINPUTS.BAS
'OARC Project 2008

'VE3KL

'Digital Input Demo

'Key to making an Iambic Keyer or any switch closure application

; See the Pinout page in the Picaxe 28X1 manual

; Output 0 to 7 are reserved for digital outputs

; In 0 to 7 are used for digital inputs

; ADC 0 to 3 are used for analog inputs

; Note what pins these refer to on the stack circuit board 



main:

if pin0=0 then	'pin0 on Picaxe is Pin 8 on the stack. zero if grounded

high 1		'turn on led or buzzer

endif

goto main










_1306485092/KEYBOARDSCANCODE.BAS
; KeyboardScanCode.bas

; Gets the keyboard scan code

; press a key on the external keyboard





main:

	kbin[1000,main],b1

	debug

	goto main




















_1306485088/SendMorseA.bas
' Send Morse A using a subroutines for dot and dash

' connect buzzer or speaker to pin 1 of picaxe to hear Morse A

' 

'

' 

'

'''''''''''''''''''''''''''''''''''

	'Initialize variabless



' 		Time =1200/WPM

' 		Time is the time of a dot in milliseconds

w0=150	'dot time    8 WPM   so it can be seen on an LED

w1=3*w0 	'Dash Time



'''''''''''''''''''''''''''''''''''''''''''''''''



' 	 Main Program

Main:

gosub MorseA

pause 1000

goto Main		'cycle forever	



'''''''''''''''''''''''''''''''''''''''''''''''''



MorseA:  		 'Light green LED 

gosub Dot

gosub Dash

pause w1

return



''''''''''''''''''''''''''''''''''''''''''''''''''''





Dot:

high 1

pause w0

low 1

pause w1

return



'''''''''''''''''''''''''''''''''''''''''''''''''''



Dash:

high 1

pause w1

low 1

pause w0

return



''''''''''''''''''''''''''''''''''''''''''''''''''''



	




















_1306485089/SET_CLOCK_1.BAS
;	Program to set the clock

;

;

	i2cslave %11010000, i2cslow, i2cbyte ;must set the correct slave address for all devices

; 	notice that I started the clock at 45 seconds, 59 mins, 23 hours, day 5 of the week,December 31,2009

;	look at the debug values flip over.





	symbol Sseconds =$00		;Start seconds

	symbol Smins=$43			;Start minutes

	symbol Shour = $19		;Start hour

	symbol Sday= $02			;Start day of the week

	symbol Sdate = $16		;Start date

	symbol Smonth = $03		;Start month

	symbol Syear = $09		;Start year

	symbol control=%00010000

;

	writei2c 0,(Sseconds,Smins,Shour,Sday,Sdate,Smonth,Syear,control)





;	Now Read the Clock



	symbol Seconds=b0

	symbol Mins=b1

	symbol Hour=b2

	symbol Day=b3

	symbol Date = b4

	symbol Month = b5

	symbol Year = b6

	

main:

	readi2c 0,(Seconds,Mins,Hour,Day,Date,Month,Year)	

	debug 

	pause 1000

goto main
























_1306485087/READ_CLOCK_1.BAS




;Read and Display Time

;Picaxe 28X1

;Sends two lines of data to the i2c display

; pause 30 is used after each write to display as it takes time for the display to process data

; first set slave to clock and then read the clock

; second set slave to display and then write to the display

; Then loop forever

; Hours Mins Seconds

; Day Month Year









	symbol Seconds=b0					;define symbols for easy reading

	symbol Mins=b1

	symbol Hour=b2

	symbol Day=b3

	symbol Date = b4

	symbol Month = b5

	symbol Year = b6





main:



	i2cslave %11010000, i2cslow, i2cbyte 	‘ set slave to clock



									‘ wait 1 sec

	readi2c 0, (Seconds, Mins, Hour, Day, Date, Month,Year) 	‘ read clock

	

	bcdtoascii Seconds,b8,b9						;get individual chrs to display		

	bcdtoascii Mins,b10,b11		;min	

	bcdtoascii Hour,b12,b13		;hour	

	bcdtoascii Date,b14,b15		;Date

	bcdtoascii Month,b16,b17	;Month

	bcdtoascii Year,b18,b19		;Year					

		

	



	i2cslave $C6,i2cslow,i2cbyte 			;set up slave to lcd  %11000110



	pause 30 						;pause 30 ms

	writei2c 0,(254,128,255)			;move to line1 position 1

	pause 30						

	

	writei2c 0,(b12,b13,":",b10,b11,":",b8,b9," H",":","M",":","S",255)

	pause 30

	writei2c 0,(254,192,255)

	pause 30						;move to line 2 position 1

	writei2c 0,(b14,b15,":",b16,b17,":",b18,b19," D",":","M",":","Y",255)

	pause 30

	

	Debug

	goto main 			




































_1306485083/OARC_SERIAL_IN_QUALIFIERS.BAS
;Serial Port Tests

;OARC Club Project 2008



;uses REALTERM a free terminal program for your PC

;This program demonstrates serial in qualifiers

;Qualifiers are often used to synchronize receivers as they acquire acquisition.

;We negelct all data until something like your call sign is received correctly.

;The program also demonstrates ASCII numbers R and S.

; Input VE3RACA into realterm send box to see what happens.

; Read the serin command in the reference manual



Main:

serin 1,T2400_4,("VE3RAC"),b1





high 1

pause 500

serout 1,T2400_4,(10,13,"> ")





if b1="R" then

serout 1,T2400_4,(b1)

b1=b1+1	;adds 1 to b1 to display ASCII S after R separated by a NULL

serout 1,T2400_4,("",b1)

high 0

pause 1000

low 0



elseif b1="A" then 

gosub MorseA



else

serout 1,T2400_4,("  Please enter VE3RACR or VE3RACA")

endif



goto Main





MorseA:

serout 1,T2400_4,("  Put your Morse A code in MorseA subroutine")

return


















_1306485085/OARCSERTXTREALTERM.BAS
;OARC Project 2009 Picaxe Group 1

;SERTXD tests.....Serial output from Picaxe to PC.

;using REALTERM  a free terminal program

;select port....in my case it is port 6. You might use a different port.

;Set RealTerm to 4800 baud then change

;Related Commands (See Picaxe command manual)

;	serrxd

;	serout...general output to an output pin such as 0;stack pin 0

;	serin....general serial input to a pin such as 1;stack pin 9

;

main:

	for b1=225to 256							; loop from 225 to 256

	sertxd("The value of b1 is ",#b1,13,10)			; serial out to programming cable

	pause 1000								; pause 1000 ms

next b1

	sertxd ("Note that b1 = 0 when b1 set to 256")		; serial out a message

	pause 5000		;note the value of b1 for b1 = 256	; pause 5000 ms

	sertxd(13,10)							; carriage return and line feed

	

	sertxd (13,10)

	sertxd (13,10)



goto main

	






_1306485055/DigitalGames8LEDS.bas
;	Digital Output  the linear way to program and a loop

;	Digital Output Pins (0,1,2,3,4,5,6,7) on the Picaxe

;	Demonstrates output, if and for statements

	let pins = 255    ;%11111111  $FF

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

	pause 3000		;pause 3000 ms

	let pins =0

	pause 3000

	let pins = 1

	pause 1000

	let pins = 2

	pause 1000

	let pins = 4

	pause 1000

	let pins = 8

	Pause 1000

	let pins = 16

	Pause 1000

	let pins = 32

	Pause 1000

	let pins = 64

	Pause 1000

	let pins = 128

	Pause 1000

	let pins = 0



;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;

	pause 3000



;now lets use a loop

	b1 =1

	for b0 = 1 to 8

	let pins = b1

	b1=b1*2

	pause 1000

	next b0

	let pins=0

	pause 3000



	let pins = %11000010	;turn some lights on using binary..more intuitive

	pause 3000

	let pins =%11111111



gosub Flash_LEDS



main:

pause 250

let pins = 195

pause 250

let pins =0

goto main

end



;now do the same using binary numbers for all situations

;do again using Hex

; you can also work with single pins

; high 0 turns on diode 1.......try it



Flash_LEDS:

for b2 = 1 to 255

let pins = b2

if b2 = 195 then

pause 10000

endif

next b2

return

end










